

Approfondissement géogébra

Préparer des visuels interactifs

I. CREER UN OUTIL

Créer un outil pas à pas

Les commandes par défaut de géogébra ne permettent pas certaines démarches évidentes pour les élèves. Ainsi il n'est pas possible de créer un carré avec un point et une longueur. Pour y remédier on peut créer de nouveaux outils.

Dans une nouvelle feuille...

Placer un point A. Taper **côté** = 4 dans la barre de saisie.

Créer un segment de longueur donnée, la longueur étant **côté**, (le nom de la variable pas sa valeur¹)

Créer un polygone régulier à quatre côtés à partir de A et B.

Créer les segments [BC], [CD] et [DA].

Dans le menu outils choisir « créer un nouvel outil ».

Les objets finaux sont B, C, D, le polygone et les segments [AB], [BC], [CD] et [DA]. (Repérer le nom des segments dans la fenêtre algèbre).

➤ *A vous de voir si vous voulez tout, vous pouvez ne prendre que les segments ou que le polygone...*

Les outils initiaux sont A et côté. (Attention l'ordre est important !)

L'outil s'appelle Carré, l'aide est « Point (crée ou non) puis longueur du côté »

Tester !

Note : cela marche bien en ligne.

Sauvegarde et utilisation : Les outils personnalisés étant créés, effacer tout et stocker le document avec un nom explicite (exemple « Vide avec outil carré »). Ce fichier sera alors stocké précieusement dans un dossier du type « public » sur le kwartz, afin que chacun puisse l'utiliser comme modèle. (Dans la gestion des outils on peut certes enregistrer les outils et ensuite le recharger mais l'ouverture d'un fichier outil oblige la fermeture du fichier courant)

II. SOIGNER LA PRESENTATION

Ecrire un texte en latex

Accrocher un texte à une courbe

Dans une nouvelle page taper dans la ligne de saisie la fonction définie par $f(x) = \frac{x^2 - 3}{x + 1}$.

On voudrait afficher l'expression de la fonction en même temps que la fonction.

Pour cela on peut choisir les propriétés de la courbe d'afficher l'étiquette, que cette étiquette soit la légende et taper l'expression de la fonction dans la légende. Essayer.

Alors heureux (se) ?

On va plutôt créer une zone de texte. Taper « **f(x)=** » cocher la case LaTeX et ajouter l'objet f. Le résultat est satisfaisant mais n'est pas lié à la courbe.

Modifier la taille, la couleur... pour que cela vous convienne totalement.

Pour l'attacher à la courbe, on crée d'abord un point sur la courbe.

Puis on prend les propriétés de la boîte de texte et dans l'onglet position, on sélectionne le point.

Le texte est maintenant lié au point, il bouge avec lui mais sa position autour du point peut encore être adaptée au besoin.

Reste à choisir la bonne place du texte et à masquer le point.

Redéfinir la fonction dans la ligne de saisie...

Noter que de choisir comme option d'étiquette nom + valeur affiche la fonction en LaTeX mais que vous n'avez pas de latitude pour déplacer cette étiquette.

Application : créer un vecteur $\overrightarrow{AB}$ affichant fièrement son nom sur la flèche.

¹ Si le nombre est tapé et pas la variable, la variable n'apparaît pas dans les outils initiaux.

Approfondissement géogébra

Préparer des visuels interactifs

III. DEVOILER UN RESULTAT PAS A PAS

Créer et utiliser des listes (1)

Afficher un texte en tableau

Afficher ligne à ligne un tableau

Dans une nouvelle page taper dans la ligne de saisie : $L1 = \{2x + 3, "\geq", 4 - 5x\}$

Vous venez de créer une liste comme vous l'indique le volet algèbre.

Créez de même les listes suivantes, dans l'ordre

$L2 = \{ "+5x", "", "+5x" \}$

$L3 = \{ 7x + 3, "\geq", 4 \}$

$L4 = \{ "-3", "", "-3" \}$

$L5 = \{ 7x, "\geq", 1 \}$

$L6 = \{ "\frac{7x}{7}", "\geq", "1/7" \}$

$L7 = \{ x, "\geq", 0.14286 \}$

Créer enfin l'anneau les contrôlant tous la liste les contenant toutes :

Liste8={L1, L2, L3, L4, L5, L6, L7}

On affiche ensuite les listes dans un tableau (Attention à la casse si vous ne faites pas de copier coller)

Tableau[Liste8, "_001001000"]

_001001000 indique qu'il y aura un trait horizontal avant les lignes 3 et 6.

Pour les traits verticaux ajouter par exemple, "|1001", (Tableau[Liste8, "_001001000|1001"])
(plus ou moins de bit que nécessaire n'est pas un problème.)

Ajouter r, c ou l entre les nombres pour l'alignement

Créer un curseur nommé n et défini par valeurs entières de 1 à 7.

Puis redéfinir le texte 1 pour qu'il dévoile n lignes en fonction du curseur

texte1=Tableau[Extraite[Liste8, 1, n], "_001001000"]

Source : Aide en ligne

Réaliser un tableau de signe avec les listes suivantes :

$\{ "x", "-\infty", "", "-3", "", "2", "", "\infty" \}$

$\{ "x-2", "", "-", " |", "-", "0", "+", "" \}$

$\{ "x+3", "", "-", " 0", "+", " |", "+", "" \}$

$\{ "f(x)", "", "+", " 0", "-", "0", "+", "" \}$

Créer un curseur et modifier votre travail pour faire apparaitre les colonnes une après l'autre. Si vous n'y arrivez pas continuez des exemples suivent.

IV. MASQUER ET AFFICHER

Dans une nouvelle page créer un segment [AB], un curseur r prenant des valeurs de 0 à 10, deux cercles de centres A et B et de rayon r, les points d'intersection de ces deux cercles et la droite qui les relie. Créer deux cases à cocher, l'un pour masquer le curseur, l'autre pour masquer les traits de construction de la médiatrice.

Les cases à cocher sont des booléens. Lorsqu'on les définit, il faut déterminer le message et les différents objets qui seront affichés/masqués.

La difficulté est qu'une fois créée, la case à cocher n'affiche jamais quels sont les objets liés et il n'est donc pas possible de les modifier ainsi.

Pour modifier les liens il faut aller dans les propriétés des objets concernés, onglet avancé. La condition pour afficher l'objet est le nom du booléen associé.

Note : il est possible de taper une formule logique du type b$\wedge$c$\vee$d

Approfondissement géogébra

Préparer des visuels interactifs

V. TABLEAU DE VARIATION

Utiliser les booléens et les cases à cocher

Utiliser les listes (2)

Programmer un script par actualisation.

A. Observation

Ouvrir le fichier <https://www.geogebra.org/m/xwdh4nfd>

Déplacer le curseur pour parcourir la courbe.

En présentation avec une classe, on s'arrête aux extrémums et on les affiche.

Après avoir parcouru ainsi la courbe remettre le curseur au minimum et cocher la case afficher le tableau.

Parcourir de nouveau la courbe et regarder la construction du tableau étape par étape.

B. Balayage vertical du tableau de variations

Ici on extrait des listes auxiliaires (liste1A et liste2A) avant de faire un tableau avec.

C. Programmation

Le texte variable lié au point et l'apparition du tableau de variation relève de la programmation de script géogébra que nous allons ensuite étudier. Il ne s'agit ici que d'une prise de contact.

Pour voir le code, afficher les propriétés du curseur, onglet script, sous onglet « par actualisation »

Un script écrit ici est exécuté chaque fois que la valeur du nombre est modifiée.

Le code présenté ici s'appuie sur la fonction dérivée de f , qui est définie sous le nom f_1 et est masquée. b est la valeur du curseur.

```
SoitValeur[texte3,Si[f_1(b)>0,LaTeX["\rotatebox{65} {Croissant}"],LaTeX["\rotatebox{-45} {Décroissant}"]]]
```

```
SoitValeur[texte3,Si[f_1(b)==0,LaTeX["C'est \; plat"],texte3]]
```

```
SoitValeur[n,Si[b<-3.5,1,Si[b<k',2,Si[b==k',3,Si[b<l',4,Si[b==l',5,7]]]]]]
```

La première ligne modifie le texte 3 selon la dérivée est positive ou non.

Pour les options de présentation du texte offerte par LaTeX dans géogébra, voir la page :

https://www.geogebra.org/manual/fr/Tutoriel:LaTeX-aide_Polices_Boites_math

La seconde gère le cas où la dérivée est nulle. La gestion se fait à part pour ne pas avoir d'instructions imbriquées. Si la dérivée n'est pas nulle le texte n'est pas modifié.

La troisième ligne découvre le tableau de signe, étape par étape. Ici des « si » imbriqués ont été utilisés. k' est l' sont les abscisses des extrémums ($k'=X[A]$, $l'=X[B]$)

D. Application

Ouvrir le fichier tableau de signe : <https://www.geogebra.org/m/cfdeckwp>

et le modifier pour que le tableau puisse lui aussi être découvert en parcourant la courbe.

E. Pour aller plus loin

Créez un tableau de signe affiché pas à pas lorsque l'on parcourt la courbe, d'une fonction affine dont le coefficient directeur et l'ordonnée à l'origine peuvent être modifiés.